
16

FIŞIERE

16.1 Fişiere, articole, identificatori

Fişierul este o structură de date contiguă. Fişierul este format din articole. Un articol grupează informaţiile referitoare la un element al unei colectivităţi omogene. De exemplu, mulţimii salariaţilor unei întreprinderi îi corespunde fişierul FISSAL.DAT.

Un articol descrie date ale unui salariat, figura 16.1

Marca
Nume
Salariu

Figura 16.1 – Structura articolului asociat mulţimii salariaţilor

Acestui şablon îi corespunde tipul derivat:

Salariat STRUC


Marca  dw ?


Nume  db 20 dup(?)


Salariu
dd ?


_sters db ?

ENDS

Câmpul _sters este folosit pentru ştergerea logică a unui articol.

Un fişier se identifică printr-un nume extern, nume utilizat la creare, deschidere, ştergere fişier şi redenumire fişier.

Identificatorul de fişer - handler - este un număr întreg pozitiv care identifică în mod unic un fişier pe disc sau alte dispozitive.  Identificatorii de fişier sunt returnaţi de funcţiile de creare şi deschidere fişier şi sunt utilizaţi la citirea din fişier, scrierea în fişier, poziţionarea în fişier şi închiderea fişierului. Sunt predefinite următoarele handlere asociate unor dispozitive standard:

Tabelul 16.1 Identificatorii unor dispozitive standard

ID
Denumire
Descriere

0
stdin
Intrare standard

1
stdout
Ieşire standard

2
stderr
Ieşire standard pentru mesaje de eroare

3
stdaux
Dispozitiv auxiliar

4
stdprn
Imprimantă standard

Fiecare fişier are asociată o etichetă în directorul părinte. Eticheta fişierului are rol de a delimita un fişier de celelalte. Ea are structura prezentată în tabelul 16.2.

Tabelul 16.2 Structura etichetei fişierelor

Octeţi
Lungime
Semnificaţie

00h - 07h
8
Nume fişier

08h - 0Ah
3
Extensie fişier

0Bh
1
Atribute fişier (Normal, RO,H,S)

0Ch - 15h
10
Rezervaţi

16h - 17h
2
Ora ultimei scrieri în fişier

18h -19h
2
Data ultimei scrieri în fişier

1Ah - 1Bh
2
Adresa de început în FAT

1Ch - 1Fh
4
Dimensiune fişier

Atributele unui fişier sunt:

0h - fişier normal

1h- fişier protejat la scriere

2h - fişier ascuns

4h - fişier de sistem

8h - identificator de volum

10h - subdirector

20h - fişier de tip arhivă.

După fiecare citire sau scriere, indicatorul curent din fişier se deplasează corespunzător cu numărul de octeţi scrişi/citiţi. După ultimul articol se scrie sfârşitul fişierului (EOF).

Lungimea fişierului de exprimă fie ca număr de octeţi fie ca număr de articole.

16.2. Operaţii cu fişiere

Pentru operaţiile cu fişiere se utilizează funcţiile DOS specifice. Principalele funcţii DOS pentru lucrul cu fişiere sunt prezentate în tabelul 16.3.

Tabelul 16.3. Funcţii DOS pentru fişiere

Cod funcţie (AH)
Descriere

3Ch
Creare fişier

3Dh
Deschidere fişier

3Eh
Închidere fişier

3Fh
Citire din fişier

40h
Scriere în fişier

41h
Ştergere fişier

42h
Poziţionare în fişier

56h
Redenumire fişier

Folosind aceste funcţii se creează macrodefiniţii pentru operaţii cu fişiere. Pentru fiecare macrodefiniţie se specifică parametrii de intrare şi cei de ieşire.

Macrodefiniţia pentru creare fişier

Parametrii de intrare:

NumeF: şir ASCIIZ ce conţine numele fişierului

Atrib: atributele de creare ale noului fişier:

- fişier protejat la scriere (1)

- fişier ascuns (2)

- fişier de sistem (4)

- fişier normal (0)

Parametrii de ieşire:

Handler: identificatorul de fişier returnat sau -1 la eroare

Rez: rezultatul operaţiei: 0 succes, altfel codul de eroare

CreeazaFisier  MACRO NumeF,Atrib,Handler,Rez


LOCAL NOK,Iesire


push ax


push cx


push dx


push ds


mov ah,3ch


mov cx,Atrib


lea dx,NumeF


int 21h


jc NOK


mov HANDLER,ax


mov Rez,0


jmp Iesire

NOK:


mov Handler,-1


mov Rez,ax

Iesire:


pop ds


pop dx


pop cx


pop ax

ENDM

Macrodefiniţia pentru deschidere fişier

Parametrii de intrare:

NumeF: şir ASCIIZ ce conţine numele fişierului

Acces: tipul de acces

0- citire

1- scriere

2- citire/scriere

Parametrii de ieşire:

Handler: identificatorul de fişier returnat sau -1 la eroare

Rez: rezultatul operaţiei: 0 succes, altfel codul de eroare
DeschideFisier MACRO NumeF,Acces,Handler,Rez


LOCAL NOK,Iesire


push ax


push dx


push ds


mov ah,3dh


mov al,Acces


lea dx,NumeF


int 21h


jc NOK


mov Handler,ax


mov Rez,0


jmp Iesire

NOK:


mov Handler,-1


mov Rez,ax

Iesire:


pop ds


pop dx


pop ax

ENDM

Macrodefiniţia pentru închidere fişier

Parametrii de intrare:

Handler: identificatorul de fişier returnat de funcţia de creare/deschidere

Parametrii de ieşire:

Rez: rezultatul operaţiei: 0 succes, altfel codul de eroare
InchideFisier MACRO Handler,Rez


LOCAL NOK,Iesire


push ax


push bx


mov ah,3eh


mov bx,Handler


int 21h


jc NOK


mov Rez,0


jmp Iesire

NOK:


mov Rez,ax

Iesire:


pop bx


pop ax

ENDM

Macrodefiniţia pentru scriere în fişier

Parametrii de intrare:

Handler: identificatorul de fişier returnat de funcţia de creare/deschidere

Buffer: zona de memorie din care se vor scrie datele citite

OctetiDeScris: numărul de octeţi de scris

Parametrii de ieşire:

OctetiScrisi: numărul de octeţi scrişi

Rez: rezultatul operaţiei: 0 succes, altfel codul de eroare
ScrieInFisier MACRO Handler,Buffer,OctetiDeScris,OctetiScrisi,Rez


LOCAL NOK,Iesire


push ax


push bx


push cx


push dx


mov ah,40h


mov bx,Handler


lea dx,Buffer


mov cx,OctetiDeScris


int 21h


jc NOK


mov OctetiScrisi,ax


mov Rez,0


jmp Iesire

NOK:


;in ax este codul de eroare!


mov Rez,ax

Iesire:


pop dx


pop cx


pop bx


pop ax

ENDM

Macrodefiniţia pentru citire din fişier

Parametrii de intrare:

Handler: identificatorul de fişier returnat de funcţia de creare/deschidere

Buffer: zona de memorie în care se vor scrie datele

OctetiDeCitit: numărul de octeţi de citit

Parametrii de ieşire:

OctetiScris: numărul de octeţi scrişi

Rez: rezultatul operaţiei: 0 succes, altfel codul de eroare
CitesteDinFisier MACRO Handler,Buffer,OctetiDeCitit,OctetiCititi,Rez


LOCAL NOK,Iesire


push ax


push bx


push cx


push dx


mov ah,3fh


mov bx,Handler


lea dx,Buffer


mov cx,OctetiDeCitit


int 21h


jc NOK


mov OctetiCititi,ax


mov Rez,0


jmp Iesire

NOK:


;in ax este codul de eroare!


mov Rez,ax

Iesire:


pop dx


pop cx


pop bx


pop ax

ENDM

Macrodefiniţia pentru ştergere fişier

Parametrii de intrare:

NumeF: şir ASCIIZ ce conţine numele fişierului

Parametrii de ieşire:

Rez: rezultatul operaţiei: 0 succes, altfel codul de eroare
StergeFisier MACRO NumeF,Rez


LOCAL NOK,Iesire


push ax


push dx


push ds


mov ah,41h


lea dx,NumeF


int 21h


jc NOK


mov Rez,0


jmp Iesire

NOK:


mov Rez,ax

Iesire:


pop ds


pop dx


pop ax

ENDM

Macrodefiniţia pentru poziţionare în fişier

Parametrii de intrare:

Handler: identificatorul de fişier returnat de funcţia de creare/deschidere

Poziţie: numărul de octeţi faţă de punctul de referinţă (început, curent, sfârşit)

From: punctul de referinţă:


0 - faţă de începutul fişierului

1- faţă de poziţia curentă din fişier

2- faţă de sfârşitul fişierului

Parametrii de ieşire:

PozCrt: noua poziţie în fişier

Rez: rezultatul operaţiei: 0 succes, altfel codul de eroare
PozitionareInFisier MACRO Handler,Pozitie,From,PozCrt,Rez


LOCAL NOK, Iesire


push ax


push bx


push cx


push dx


mov ah,42h


mov al,From


mov bx,Handler


mov dx,word ptr Pozitie[0]


mov cx,word ptr Pozitie[1]


int 21h


jc NOK


mov word ptr PozCrt[0],ax


mov word ptr PozCrt[1],dx


mov Rez,0


jmp Iesire

NOK:


mov Rez,ax

Iesire:


pop dx


pop cx


pop bx


pop ax

ENDM
Deoarece sistemul de operare Windows 95 permite utilizarea fişierelor cu nume lungi (maxim 256 caractere), pentru acestea există funcţii specifice. În tabelul 16.4 sunt prezentate acestea.

Tabelul 16.4 Funcţii specifice fişierelor sub Windows 95

Cod funcţie (AX)
Descriere

7141h
Ştergere fişier

7156h
Redenumire fişier

716Ch
Creare/Deschidere fişier

16.3 Utilizarea de fişiere pentru calculul fondului de salarii

Se consideră structura unui articol prezentată în paragraful 16.1, ce conţine caracteristicile salariaţilor. Utilizând macrodefiniţiile prezentate anterior se scrie un program ce realizează următoarele funcţii:

· adăugare de articole în fişier

· modificarea câmpului salariu al unui articol identificat după marcă

· ştergerea logică a unui articol

· calculul fondului de salarii

Programul afişează un meniu ce permite alegerea unei operaţii.

include fmacros.mac

include conv.mac

dosseg


.model small


.stack 100h


.data

;structura persoana

pers
struc


marca  dw ?


nume
db 20 dup(?)


salariu dw ?


_sters db 0

pers
ends

;lungimea articolului

lart
equ TYPE pers

;buffere pentru memorarea numerelor

smrc
db  6 dup (?)

ssal
db  6 dup (?)

;variabila temporara pentru marca

mrc
dw ?

OctCit
dw ?

OctScr dw ?

CRLF db 13,10,'$'

MsjNume db 'Nume salariat:$'

MsjMarca db 'Marca:$'

MsjSal db 'Salariu [mii lei]:$'

MsjSalOld db 'Vechiul salariul [mii lei]:$'

Mesaj
db 'Articol inexistent!',13,10,'$'

MsjSterg db 'Se sterge acest articol? [0/1]->$'

MsjRefac db 'Articol este sters. Se reface? [0/1]->$'

sir_adaug
db '[1] Adaugare',10,13,'$'

sir_modif
db '[2] Modificare',10,13,'$'

sir_sterg
db '[3] Stergere',10,13,'$'

sir_calcul
db '[4] Calcul',10,13,'$'

sir_ies
db '[5] Iesire',10,13,'$'

sir_opt
db '-->$'

;numele fisierului

nf
db 'fissal.dat',0

;identificator fisier

h
dw ?

;baza de numarare

baza dw 10

;articolul utilizat la citire si scriere

om pers <>

;variabile utilizate in program

car
db ?

er
dw ?

modp
db ?

poz
dd ?

unde
dd ?


.code

start:


mov ax,@data


mov ds,ax

meniu:


StergeEcran


afis sir_adaug


afis sir_modif


afis sir_sterg


afis sir_calcul


afis sir_ies


afis sir_opt


mov ah,01h


int 21h


cmp al,'1'


jne Pas1


jmp MAdaug

Pas1:


cmp al,'2'


jne Pas2


jmp MModif

Pas2:


cmp al,'3'


jne Pas3


jmp MSterg

Pas3:


cmp al,'4'


jne Pas4


jmp MCalcul

Pas4:


cmp al,'5'


jne Pas5


jmp ies_dos

Pas5:


;s-a tastat altceva


jmp meniu

;S-a ales optiunea 1, adaugarea unui salariat

;Pasi:

;Se citeste marca

;Se citeste numele

;Se citeste salariul

;Se deschide fisierul, se face pozitionarea la sfirsit

;si se scrie articolul

MAdaug:


afis CRLF


;se citeste marca


afis MsjMarca


citsn smrc


atoi smrc,om.marca,baza


;se citeste numele


afis MsjNume


cit  om.nume


;se citeste salariul


afis MsjSal


citsn ssal


atoi ssal,om.salariu,baza


;deschidere fisier pt scriere


DeschideFisier nf,1,h,er


cmp er,0


je _adauga


cmp er,2 ;file not found


je CreeazaF


jmp end_A

CreeazaF:


;fisierul nu exista. Se creeaza un fisier normal


CreeazaFisier nf,0,h,er


cmp er,0 ;


je _adauga


jmp end_A

_adauga:


;pozitionare la sfirsitul fisierului


mov word ptr poz[0],0 ;!!!


mov word ptr poz[1],0 ;0 octeti


mov modp,2 ;...fata de sfirsit


PozitionareInFisier h,poz,modp,unde,er


cmp Er,0


je Scrie


jmp end_A

Scrie:


ScrieInFisier h,om,lart,OctScr,er


InchideFisier h,er

end_A:


jmp meniu

;S-a ales optiunea 2, modificarea salariului

;Pasi:

;Se citeste marca

;Se parcurge secvential fisierul pina cind se gaseste ;marca respectiva

;Daca se gaseste dar articolul este sters se iese cu mesaj

;Daca se ajunge la sfirsit de fisier se iese cu mesaj

;Daca s-a gasit articolul, se citeste noul salariu

;si se actualizeaza in fisier, dupa ce in prealabil

;s-a facut deplasarea corespunzatoare in fisier

MModif:


afis CRLF


afis MsjMarca


citsn smrc


atoi smrc,mrc,baza


DeschideFisier nf,2,h,er


mov cx,0

mloop:


CitesteDinFisier h,om,lart,OctCit,er


inc cx


cmp OctCit,0 ;s-a ajuns la sfirsit?


jne next_1


jmp sf_mloop_1

next_1:


mov ax,mrc


cmp om.marca,ax


je modi


jmp mloop

modi:


cmp om._sters,1


jne next_2


jmp sf_mloop_1

next_2:


afis CRLF


afis MsjSalOld


afisn om.salariu,baza


afis CRLF


afis MsjSal


citsn ssal


atoi ssal,om.salariu,baza


dec cx


mov ax,cx


mov cx,lart


mul cx


mov word ptr poz[0],ax


mov word ptr poz[1],0


mov modp,0


PozitionareInFisier h,poz,modp,unde,er


ScrieInFisier h,om,lart,OctScr,er


jmp sf_mloop


;nu s-a gasit articolul cu marca respectiva

sf_mloop_1:


afis Mesaj


citc al

sf_mloop:


InchideFisier h,er


jmp meniu

;S-a ales optiunea 3, stergerea logica a unui articol

;Pasi:

;Se citeste marca

;Se parcurge secvential fisierul pina cind se gaseste ;marca respectiva

;Daca se gaseste si articolul este sters se

;Daca se ajunge la sfirsit de fisier se iese cu mesaj

;Daca s-a gasit articolul si nu este sters, este posibila

;reactualizarea articolului. In caz negativ se iese.

;Se afiseaza datele despre salariat si ce cere confirmarea

;de stergere

;Se actualizeaza fisierul in cazul unui raspuns afirmativ ;pentru

;ambele situatii

MSterg:


;se citeste marca


afis CRLF


afis MsjMarca


citsn smrc


atoi smrc,mrc,baza


DeschideFisier nf,2,h,er


mov cx,0 ;cx memoreaza numarul de articole citite

sloop:


CitesteDinFisier h,om,lart,OctCit,er


inc cx


cmp OctCit,0 ;s-a ajuns la sfirsit?


jne snext_1


jmp sf_sloop_1

snext_1:


;s-a gasit articolul cautat?


mov ax,mrc


cmp om.marca,ax


je sterg


jmp sloop

sterg:


afis CRLF


afis MsjNume


afis om.nume


afis CRLF


afis MsjSal


afisn om.salariu,baza


afis CRLF


cmp om._sters,1


jne snext_2


;este sters. intreb daca-l refac


afis MsjRefac


citc car


cmp car,'1'


je crt


jmp sf_sloop

crt:


mov om._sters,0


jmp modifs

snext_2:


;intreb daca se sterge


afis MsjSterg


citc car


cmp car,'1'


je crt1


jmp sf_sloop

crt1:


mov om._sters,1

modifs:


dec cx


mov ax,cx


mov cx,lart


mul cx


mov word ptr poz[0],ax


mov word ptr poz[1],0


mov modp,0


PozitionareInFisier h,poz,modp,unde,er


ScrieInFisier h,om,lart,OctScr,er


jmp sf_sloop


;nu s-a gasit articolul cu marca respectiva

sf_sloop_1:


afis Mesaj


citc al

sf_sloop:


InchideFisier h,er


jmp meniu

;S-a ales optiunea 4, calcul fond salarii

;Pasi:

;Se deschide fisierul

;Se citesc toate articolele pina la sfirsit si salariile

;articolelor nesterse se insumeaza

;Se afiseaza rezultatul

MCalcul:


DeschideFisier nf,0,h,er


mov ax,0

cloop:


CitesteDinFisier h,om,lart,OctCit,er


cmp OctCit,0


je sf_cloop


cmp om._sters,1


jne adun


jmp cloop

adun:


add ax,om.salariu


jmp cloop

sf_cloop:


InchideFisier h,er


afis CRLF


afis MsjSal


afisn ax,baza


mov ah,01h


int 21h


jmp meniu

ies_dos:


mov ax,4c00h


int 21h

end start
unde “conv.mac” este numele fişierului ce conţine următoarele macrodefiniţii (constituite într-o bibliotecă):

;Macrodefinitie pentru citirea unui sir de numere

;citirea se opreste pana introduc orice altceva

citsn 
MACRO sir


local etr,et1


push ds


push bx


push ax


push dx


lea bx,sir

etr:


mov ah,08h


int 21h


cmp al,'9'


jg et1


cmp al,'0'


jb et1


mov [bx],al


mov dl,al


mov ah,02h


int 21h


inc bx


jmp etr

et1:


mov byte ptr[bx],'$'


mov dl,13


mov ah,02h


int 21h


mov dl,10


mov ah,02h


int 21h


pop dx


pop ax


pop bx


pop ds

ENDM

;Macrodefinitie pentru afisarea unui sir de caractere

afis 
macro sir


push ds


push dx


push ax


lea dx,sir


mov ah,09h


int 21h


pop ax


pop dx


pop ds

endm

;Macrodefinitie pentru conversie ASCII-binar

;parametri: numar: DB, num: DWORD, baza: DWORD

;numarul se afla in ax

atoi
MACRO numar,num,baza


local etsf,rev


push ds


push si


push ax


push bx


lea si,numar


xor ax,ax


xor bx,bx

rev:


mov bl,[si]


cmp  bl,'$'


jz etsf


sub bl,30h


mul baza


add ax,bx


inc si


jmp rev

etsf:


mov num,ax


pop bx


pop ax


pop si


pop di

ENDM

;Macrodefinitie pentru conversie binar-ASCII

;si afisare

;numarul este preluat din 'nr'

afisn 
MACRO nr,baza


local rel,etc,etsf


push bp


push sp


push ax


push dx


mov bp,sp


mov ax,nr

rel:


xor dx,dx


or ax,0


jz etc


div baza


add dx,30h


push dx


jmp rel

etc:


cmp bp,sp


jz etsf


pop dx


mov ah,02h


int 21h


jmp etc

etsf:


pop dx


pop ax


pop sp


pop bp

ENDM

;macro pt. citirea unui caracter de la tastatura

citc 
macro car


push ax


mov ah,01h


int 21h


mov car,al


pop ax

endm

cit MACRO buf


local et1,et2



push ax


push cx


push bx


lea bx,buf


mov cx,20


xor si,si

et1:


jcxz et2


mov ah,01h


int 21h


cmp al,13


je et2


mov ah,0


mov bx[si],ax


dec cx


inc si


jmp et1

et2:


mov bx[si],word ptr '$'


mov dl,13


mov ah,02h


int 21h


mov dl,10


mov ah,02h


int 21h


pop bx


pop cx


pop ax

ENDM

StergeEcran MACRO


push ax


mov al,02h


mov ah,00h


int 10h


pop ax

ENDM

Structura de selecţie a opţiunilor din meniu, precum şi alte secvenţe de program în care se iau decizii s-au construit în această manieră deoarece prin expandarea macrodefiniţiilor în program, salturile condiţionale (jne, je), depăşesc limita de 127 octeţi.

Se observă că dimensiunea variabilei registru în care se însumează salariile este de tip cuvânt (DW), în mod real fiind insuficientă pentru calculul  salariilor exprimate în mii lei, pentru un număr mai mare de articole.

16.4 Operaţii cu directori şi subdirectori

Operaţiile realizate asupra [sub]directorilor sunt: crearea, redenumirea, ştergerea şi schimbarea directorului curent. Pentru aceasta se utilizează următoarele funcţii DOS din tabelul 16.5. 

Tabelul 16.5. Funcţii pentru directori

Cod funcţie (AH)
Descriere

39h
Creare director

3Ah
Ştergere director

3Bh
Setare director curent

În tabelul 16.6 sunt prezentate funcţiile echivalente specifice sistemului de operare Windows 95, pentru nume lungi de fişiere. Parametrii acestor funcţii rămânând neschimbaţi.

Tabelul 16.6. Funcţii pentru directori specifice Windows 95

Cod funcţie (AX)
Descriere

7139h
Creare director

713Ah
Ştergere director

713Bh
Setare director curent

În continuare sunt prezentate macrodefiniţiile pentru operaţii cu directori, utilizând funcţiile specifice numelor lungi de fişiere.

Macrodefiniţia pentru creare director 

Parametrii de intrare:

NumeDir: şir ASCIIZ ce conţine numele directorului ce se va crea

Parametrii de ieşire:

Rez: rezultatul operaţiei: 0 succes, altfel codul de eroare
CreeazaDirector MACRO NumeDir,Rez

  LOCAL NOK, Iesire

  push ds

  push dx

  push ax

  mov ax,7139h

  lea dx,NumeDir

  int 21h

  jc NOK

  mov Rez,0

  jmp Iesire

NOK:

  mov Rez,ax

Iesire:

  pop ax

  pop dx

  pop ds

ENDM
Macrodefiniţia pentru ştergere director

Parametrii de intrare:

NumeDir: şir ASCIIZ ce conţine numele directorului ce va fi şters

Parametrii de ieşire:

Rez: rezultatul operaţiei: 0 succes, altfel codul de eroare

StergeDirector MACRO NumeDir,Rez

  LOCAL NOK, Iesire

  push ds

  push dx

  push ax

  mov ax,713Ah

  lea dx,NumeDir

  int 21h

  jc NOK

  mov Rez,0

  jmp Iesire

NOK:

  mov Rez,ax

Iesire:

  pop ax

  pop dx

  pop ds

ENDM

Macrodefiniţia pentru schimbare director curent

Parametrii de intrare:

NumeDir: şir ASCIIZ ce conţine numele noii căi

Parametrii de ieşire:

Rez: rezultatul operaţiei: 0 succes, altfel codul de eroare 

SchimbaDirector MACRO NumeDir, Rez

  LOCAL NOK, Iesire

  push ds

  push dx

  push ax

  mov ax,713Bh

  lea dx,NumeDir

  int 21h

  jc NOK

  mov Rez, 0

  jmp Iesire

NOK:

  mov Rez,ax

Iesire:

  pop ax

  pop dx

  pop ds

ENDM
În continuare se prezintă un exemplu de utilizare a acestor macrodefiniţii, prin secvenţa de program creându-se un subdirector în directorul curent şi se va schimba directorul curent cu acesta.

.model small

.stack 100h

.data

dir db 'Subdirector cu nume lung... ', 0

cod dw ?

.code

start:

   mov ax,@data

   mov ds,ax

   CreeazaDirector dir,cod

   SchimbaDirector dir,cod

   mov ax,4c00h

   int 21h

   end start

Căutarea fişierelor în directoare se realizează cu funcţiile DOS FindFirst (AH=4Eh) şi FindNext, (AH=4Fh). Pentru Windows 95 există funcţiile corespunzătoare AX=714Eh respectiv, AX=714Fh. 

Programatorul are la dispoziţie prin funcţii DOS toate resursele pentru a implementa tehnicile şi metodele de organizare şi căutare a articolelor în fişier implementate în limbajul de programare COBOL.

